Maple in Math 647

We will be using Maple for some symbolic computations. I'll give you some of the basics in this document; of course it can get much more sophisticated than what I'm doing. I'm actually creating this document in Maple 12, and you will probably want to do the same for your homework. There are two interfaces in Maple 12: document mode and the classic worksheet. I'm using document mode. On calclab, this is the button labeled something like "Maple Java".

I'm inputting this as text (so that Maple doesn't interpret all of this as one long equation). To get to text mode, click on "text" in the upper left. Of course, we'll need to use Maple for more than a not-very-good word processor. Although you can put math right into text, I'll use the "insert Maple input" button on the tool-bar. It's the symbol which looks like what I've got below. I'll add 2 and 2; notice that I put a semicolon at the end of the input. When I hit return, it gives me another Maple execution group.

> $2+2;$

$$4 \quad \textbf{(1)}$$

>

To get back to text, I pressed the "insert plain text" button on the toolbar, which is the capital T. Using the toolbar, you can use different fonts, different sized letters, bold, italics, and so on. You can actually put live Maple in your text (this is the difference between document mode and the worksheet). Just click on math in the upper left. Here's how I could take 2 to the 20th power: 2^{20};

$$1048576 \quad \textbf{(2)}$$

I put it in Math mode, and then typed in "2^20", then right arrow to get out of the exponent, then semicolon, then return. I suggest that you don't do it this way in general; it's harder for someone else to follow when they look at what you do.

Defining and evaluating functions and expressions

One feature of Maple that takes some getting used to is that "functions" and "expressions" are similar, but not the same. First, I'll define *f* as a function.

> $f := x \rightarrow x^3 + 3\cdot x;$

$$f := x \rightarrow x^3 + 3\,x \quad \textbf{(3)}$$

What I actually typed was "f:=x->x^3+3*x". Some comments: notice that I put a colon sign and then an equals after f. Think of ":=" as saying "defined to be equal to". The "x->" is saying that this will be a function of x, and -> turns into an arrow. The "^" puts me up in the exponent. To get out of it (and type the +), I used the right arrow key. This is useful to remember: the right arrow key will get you out of a lot of stuff. I end the input line with a semicolon. (If I want to suppress output, I'd use a colon.) Now that I've got a function, let's evaluate it at 2.

> $f(2);$

$$14 \quad \textbf{(4)}$$

which is of course what it should be. Now I'll define *g* to be an expression.

> $g := x^3 + 3\cdot x;$

$$g := x^3 + 3\,x \quad \textbf{(5)}$$

To plug an x value into an expression, use "subs":

> $subs(x=2, g);$

$$14 \quad \textbf{(6)}$$

A common mistake in defining functions is to try something like "h(x):=x^3+3*x". This doesn't work: Maple gets confused and will complain.

Differentiating and integrating

In differentiating and integrating, you have to treat functions and expressions differently. First I differentiate the expression g which I defined above. One way to do it is with the "diff" command.

```
> diff(g, x);
```

$$3x^2+3 \quad \textbf{(7)}$$

Here, the "x" told it to differentiate with respect to x. I recommend another way, however. Use "Diff" (which doesn't evaluate it), and then "value" to evaluate it. This way you can check for typos.

```
> Diff(g, x); value(% );
```

$$\frac{d}{dx}\left(x^3+3x\right)$$

$$3x^2+3 \quad \textbf{(8)}$$

Notice that I put two statements on a line. That works fine, as long as you have a semicolon between them. Also notice the "%". This says to plug in the most recently executed statement. Be careful using this, since if you're jumping around in a worksheet, the most recently executed statement might not be what you think it is. The result of the above is an expression. If I want, I can give that expression a label, for example "Dg":

```
> Diff(g, x); Dg := value(% );
```

$$\frac{d}{dx}\left(x^3+3x\right)$$

$$Dg := 3x^2+3 \quad \textbf{(9)}$$

To differentiate a function, you can use "D".

```
> D(f);
```

$$x \rightarrow 3x^2+3 \quad \textbf{(10)}$$

The result is a function. If I want to find the derivative of f at 2, here's one way:

```
> Df := D(f);
```

$$Df := x \rightarrow 3x^2+3 \quad \textbf{(11)}$$

```
> Df(2);
```

$$15 \quad \textbf{(12)}$$

Using the expression Dg, I'd have to use "subs" again.

```
> subs(x = 2, Dg);
```

$$15 \quad \textbf{(13)}$$

For integration, the Maple commands "Int" and "int" expect an expression. We can turn the function f into an expression by putting in $f(x)$. Get into the habit of using "Int" and "value" rather than "int", since this lets you check for typos. Here are some definite integrals. I enter the interval [0,2] as "x=0..2". You use this a lot.

```
> Int(f(x), x = 0 ..2); value(% );
```

$$\int_0^2 \left(x^3+3x\right)\,dx$$

$$10 \quad \textbf{(14)}$$

```
> Int(g, x = 0 ..2); value(% );
```

$$\int_0^2 \left(x^3+3x\right)\,dx$$

(15)

$$10 \quad \textbf{(15)}$$

Here are some indefinite integrals. Notice that Maple doesn't add "+C".

```
> Int( f(x), x); value(% );
```

$$\int \left(x^3 + 3\,x\right) \mathrm{d}x$$

$$\frac{1}{4}\,x^4 + \frac{3}{2}\,x^2 \quad \textbf{(16)}$$

```
> Int(g, x); value(% );
```

$$\int \left(x^3 + 3\,x\right) \mathrm{d}x$$

$$\frac{1}{4}\,x^4 + \frac{3}{2}\,x^2 \quad \textbf{(17)}$$

Plotting

The "plot" command expects an expression. You must put in a range for x, you can put in a range for y.

```
> plot(g, x = 0 ..2, y = 0 ..10);
```



To plot the function f, we first have to turn it into an expression by evaluating it at x.

```
> plot( f(x), x = 0 ..2, y = 0 ..10);
```



Solving equations

Use "solve" for exact solutions.

> *sol* := *solve*(*f*(*x*) = 1, *x*);

$$sol := \frac{1}{2}\left(4+4\sqrt{5}\right)^{1/3} - \frac{2}{\left(4+4\sqrt{5}\right)^{1/3}}, -\frac{1}{4}\left(4+4\sqrt{5}\right)^{1/3} + \frac{1}{\left(4+4\sqrt{5}\right)^{1/3}} + \frac{1}{2} I\sqrt{3}\left(\frac{1}{2}\left(4+4\sqrt{5}\right)^{1/3} + \frac{2}{\left(4+4\sqrt{5}\right)^{1/3}}\right), -\frac{1}{4}\left(4+4\sqrt{5}\right)^{1/3} + \frac{1}{\left(4+4\sqrt{5}\right)^{1/3}} - \frac{1}{2} I\sqrt{3}\left(\frac{1}{2}\left(4+4\sqrt{5}\right)^{1/3} + \frac{2}{\left(4+4\sqrt{5}\right)^{1/3}}\right) \quad \textbf{(18)}$$

Pretty nasty looking. Notice the "I" in two of the roots: Maple is giving us all of the complex roots. We can get numbers out of this by using "evalf".

> *evalf*(*sol*);

$$0.3221853540, -0.1610926772 + 1.754380960\, I, -0.1610926772 - 1.754380960\, I \quad \textbf{(19)}$$

Many times you hit things which can't be solved exactly. For example, suppose that we want to find where $y = x^2$ crosses $y = \cos(7 \cdot x)$. First I'll try "solve".

> $solve\left(x^2 = \cos(7 \cdot x), x\right)$;

```
Warning, solutions may have been lost
```

Maple comes up empty. So, I'll go for numerical solutions, using "fsolve".

> $fsolve(x^2 = cos(7 \cdot x), x);$

$$0.2176307779 \quad (20)$$

Looks great. Unfortunately, there's a problem in that I've missed 5 out of the 6 solutions. It's immediately obvious when I look at a plot of the two functions. I need only plot from -1 to 1, since outside of that interval x^2 is larger than 1, and can't equal $cos(7 \cdot x)$, which is always between -1 and 1. Notice how I use curly braces to plot two expressions on one graph.

> $plot(\{x^2, cos(7 \cdot x)\}, x = -1..1);$



To get all of the roots, I have to put different intervals into fsolve. I get the intervals by looking at the plot. Get into the habit of using plots whenever possible when you solve equations, especially when you use fsolve. Here are the three negative roots.

> $root1 := fsolve(x^2 = cos(7 \cdot x), x = -1..-0.8);$

$$root1 := -0.9567758885 \quad (21)$$

> $root2 := fsolve(x^2 = cos(7 \cdot x), x = -0.8..-0.6);$

$$root2 := -0.7615922217 \quad (22)$$

> $root3 = fsolve(x^2 = cos(7 \cdot x), x = -0.4..0);$

$$root3 = -0.2176307779 \quad (23)$$

Here's an example which uses parametric curves, and also has us solve a system of equations. Let's find the intersections of the two ellipses which are parameterized as $x = 2 \cdot \sin(t), y = 3 \cdot \cos(t)$ and $x = 3 \cdot \sin(t), y = \cos(t)$. First off, I'll demonstrate how to plot a parametric curve by plotting the first one:

> $plot([2 \cdot sin(t), 3 \cdot cos(t), t = 0..2 \cdot \pi], scaling = constrained);$



Notice the placement of the square brackets in the plot command. To put two parametric curves on the same plot, I put both sets of square brackets inside curly braces:

> $plot(\{[2 \cdot sin(t), 3 \cdot cos(t), t = 0..2 \cdot \pi], [3 \cdot sin(s), cos(s), s = 0..2 \cdot \pi]\}, scaling = constrained);$



I'm using different parameters on the different curves. Now, the two curves cross at points where the *x* and *y* coordinates are equal. In other words we get the two equations

> $eq1 := 2 \cdot sin(t) = 3 \cdot sin(s);$

$$eq1 := 2\sin(t) = 3\sin(s) \quad \textbf{(24)}$$

> $eq2 := 3 \cdot cos(t) = cos(s);$

$$eq2 := 3\cos(t) = \cos(s) \quad \textbf{(25)}$$

and we want to solve these two equations in the two unknowns *s* and *t*. First I'll look for exact solutions using solve.

> $solve(\{eq1, eq2\}, \{s, t\});$

$$\left\{s = \arctan\left(\frac{4}{15}\, RootOf\left(-2 + 77\,\_Z^2, label = \_L6\right)\sqrt{5}\,\sqrt{77}\right), t = \arcsin\left(6\, RootOf\left(-2 + 77\,\_Z^2, label = \_L6\right)\right)\right\} \quad \textbf{(26)}$$

That doesn't look very good. How about using fsolve for numerical solutions? I'll look for the intersection in the first quadrant, so that *s* and *t* are both between 0 and $\frac{\pi}{2}$.

> $fsolve\left(\{eq1, eq2\}, \{s, t\}, \left\{s = 0 .. \frac{\pi}{2}, t = 0 .. \frac{\pi}{2}\right\}\right);$

$$\{s = 0.7005763163, t = 1.313131055\} \quad \textbf{(27)}$$

So, how do I plug one of these parameter values back into the parametric equations? It's crass to type them in again. Instead, I should have labeled the solution something. let's try again.

```
> solution := fsolve({eq1, eq2}, {s, t}, {s = 0 .. Pi/2, t = 0 .. Pi/2});
```

$$solution := \{s = 0.7005763163, t = 1.313131055\} \quad \textbf{(28)}$$

Now I'll let "ssol" to be the solution for s. The solution for s is the right hand side of the first term of the list "solution", so

```
> ssol := rhs(solution[1]);
```

$$ssol := 0.7005763163 \quad \textbf{(29)}$$

is how you do it. Now let's plug this into the equations for the second ellipse, and call the intersection (xsol,ysol).

```
> xsol := 3· sin(ssol);
```

$$xsol := 1.933975114 \quad \textbf{(30)}$$

```
> ysol := cos(ssol);
```

$$ysol := 0.7644707871 \quad \textbf{(31)}$$

To check this, you can find the solution for t and plug it into the equations for the first ellipse.

```
> tsol := rhs(solution[2]);
```

$$tsol := 1.313131055 \quad \textbf{(32)}$$

```
> 2· sin(tsol); 3· cos(tsol);
```

$$1.933975114$$
$$0.7644707859 \quad \textbf{(33)}$$

Not exact, but pretty darn good.

Differential equations

The command that Maple uses to solve ordinary differential equations is "dsolve". Here are some examples. Notice how I'm using t$2 in diff to take a second derivative.

```
> deq1 := diff(y(t), t$2) + y(t) = 0;
```

$$deq1 := \frac{d^2}{dt^2} y(t) + y(t) = 0 \quad \textbf{(34)}$$

```
> dsolve(deq1, y(t));
```

$$y(t) = \_C1 \sin(t) + \_C2 \cos(t) \quad \textbf{(35)}$$

Without specifying initial conditions, we got the general solution. Now let's look for the solution which satisfies the initial conditions y(0)=1 and y'(0)=2.

```
> dsolve({deq1, y(0) = 1, D(y)(0) = 2}, y(t));
```

$$y(t) = 2 \sin(t) + \cos(t) \quad \textbf{(36)}$$

Generally you can't solve a differential equation exactly. For example:

```
> deq2 := diff(y(t), t $ 2) − sin(y(t)) = 1;
```

$$deq2 := \frac{d^2}{dt^2} y(t) - \sin(y(t)) = 1 \quad \textbf{(37)}$$

```
> dsolve({deq2, y(0) = 0, D(y)(0) = 1}, y(t));
```

$$y(t) = RootOf\left(-\left(\int_0^{\_Z} \frac{1}{\sqrt{-2\cos(\_a) + 2\_a + 3}}\, d\_a\right) + t\right) \quad \textbf{(38)}$$

This is not much help. However, we can get a numerical solution. I use "output=listprocedure" so that I can get a hold the function which solves it.

> $dsol := dsolve(\{deq2, y(0) = 0, D(y)(0) = 1\}, y(t), numeric, output = listprocedure);$

$$dsol := \left[t = \textbf{proc}(t)\ \ldots\ \textbf{end proc}, y(t) = \textbf{proc}(t)\ \ldots\ \textbf{end proc}, \frac{\mathrm{d}}{\mathrm{d}t}\, y(t) = \textbf{proc}(t)\ \ldots\ \textbf{end proc}\right] \quad \textbf{(39)}$$

Looking at the solution, which I've labelled "dsol", I can see that the function y(t) that I want is the right hand side of the second equation in the list dsol. I'll call that function ysol (so that if I go back to the original equation, I don't mess things up by putting in a specific function for what should be a variable). Here's how you do it:

> $ysol := rhs(dsol[2]);$

$$ysol := \textbf{proc}(t)\ \ldots\ \textbf{end proc} \quad \textbf{(40)}$$

So, is this what we want? Let's plot it and find out.

> $plot(ysol(t), t = -2..2);$



>

At least it looks plausible: it has the right value and derivative at 0.